Derivatives, Slopes, Tangent Lines, and Making Movies

Chapter 3 Technology Application Project

1 Introduction

OBJECTIVE: To visualize the derivative and the linearization of a function at a point.

In this module, we explore the derivative as the slope of a nonlinear function and find the equation of the line tangent to a curve at a point. You will learn how to plot the curve and selected tangents on the same graph. In addition, you will see how to use Maxima to make a movie animation by generating a sequence of plots, each showing a different tangent to the curve. When the sequence of graphs is animated, the tangent lines appear to roll along the graph of the function.

1.1 Technology Guidelines

NOTE: If you have just finished a document, restart Maxima before executing a new document. This can be done by choosing "Restart" from the Maxima menu.

TO OPEN OR CLOSE CELLS
Click on the arrow at the top of the cell bracket.

TO STOP AN EXECUTION
Click on STOP button from the toolbar.

ORDER OF EXECUTION
Execute commands in the order given. Do not skip any Maxima Input lines within a given document.

Alternatively, you can execute the entire worksheet by selecting the "Evaluate All Cells" command from the "Cell" drop down menu or simply press Ctrl-r.

SAVING WORKSHEETS
You can save anytime to any directory you choose, and it is wise to save often.

EXPERIENCING MAJOR PROBLEMS
Save if appropriate, and then shut down Maxima and start it up again.

1.2 You Try It

First, work through Parts I - V with the example function f(x) = x^2, and then repeat the steps in Parts I - V for some functions that you select. Here are some suggestions.

1. x^3 for -2<=x<=2
2. sin(x) for 0<=x<=2*%pi
3. %e^(-x^2) for -2<=x<=2
4. sqrt(x) for 0<x<=4
   (Note that we do not include x=0. Do you know why?)

2 Part I: The Derivative at a Point

Section 3.1

Define a nonlinear function of your choice and call it f(x) and graph it. For an example, we choose the function f(x) = x^2 for -2<=x<=2 (To put in a different function and domain, change the entries in the following input cell.)

```
(%i1) numer:true$
      kill(all)$
      load(draw)$
      ratprint:false$
```

```
(%i3) x0: -2;
      xf: 2;
      f(x) := x^2;
(%o3) -2
(%o4) 2
(%o5) f(x):=x^2
```

```
(%i6) wxdraw2d(
        nticks=200,
        color=blue,
        explicit(f(x),x,x0,xf),
        xlabel="x",
        ylabel="y");
```

(%t6)



(%o6)

Now pick a point on the function and use the definition of the derivative to find the slope of the function's graph at the point you pick, calling it mtangent. We choose x = 1 for our example.

```
(%i7) mtangent: limit((f(1+h)-f(1))/h,h,0);
```

(%o7) 2

3 Part II: The Linearization of a Function

Sections 3.1 and 3.8 (ET Sections 3.1 and 3.10)

Form a new function for the line that is tangent to the function that you chose in the "You Try It" section at that point you picked in Part I, and call it ytangent = L(x). This function is called the linearization of y = f(x) at the point (1, f(1)).

```
(%i8) L(x) := f(1)+mtangent*(x-1);
```

(%o8) $L(x):=f(1)+mtangent\ (x-1)$

```
(%i9) wxdraw2d(
        nticks=200,
        color=red,
        explicit(f(x),x,x0,xf),
        color=blue,
        explicit(L(x),x,x0,xf),
        xlabel="x",
        ylabel="y, ytangent");
```

(%t9)



(%o9)

4 Part III: The Derivative Function

```
Section 3.1

Form the derivative function that will give the slope of the
tangent to your chosen function at any point with coordinates
(x, f(x)). Call the derivative function yprime, and graph
it.
```

```
(%i10) yprime: limit((f(x+h)-f(x))/h,h,0);
```

(%o10) $2\,x$

```
(%i11) wxdraw2d(
         nticks=200,
         color=green,
         explicit(yprime,x,x0,xf),
         xlabel="x",
         ylabel="yprime");
```

(%t11)



(%o11)

5 Part IV: A Whole Bunch of Tangents

```
Section 3.1

Form a new Maxima function that gives the equation of the line
tangent to y = f(x) at the point (a, f(a)). Call the new
function tanline(x,a).
```

```
(%i12) tanline(x,a) := f(a)+subst(a,x,yprime)*(x-a);
```

(%o12) $\text{tanline}(x,a) := f(a) + \text{subst}(a,x,yprime)\,(x-a)$

```
Test your tanline(x,a) for several values of a by plotting the
tangent lines and y = f(x) together on the same graph. First,
use the tanline(x,a) function and a short loop to generate a
list of equations for the tangents to the curve at points
(a, f(a)), for values of a varying from -2 to 2 in increments
of 0.1. Then graph the tangent lines and y = f(x) together on
the same graph.
```

```
(%i13) numer:true$
```

```
(%i14) g: []$
       listoftangents: makelist(tanline(x,x0+(xf-x0)*a/20), a, 0, 20);
       for i: 1 thru 21 do g: append(g,[explicit(listoftangents[i],x,x0,xf)]);
```

(%o15) $[4-4\,(x+2),3.24-3.6\,(x+1.8),2.56-3.2\,(x+1.6),1.96-2.8\,(x+1.4),1.44-2.4\,(x+1.2),1-2\,(x+1),0.64-1.6\,(x+0.8),0.36-1.2\,(x+0.6),0.16-0.8\,(x+0.4),0.04-0.4\,(x+0.2),0,0.4\,(x-0.2)+0.0400000000000001,0.8\,(x-0.4)+0.16,1.2\,(x-0.6)+0.36,1.6\,(x-0.8)+0.64,2\,(x-1)+1,2.4\,(x-1.2)+1.44,2.8\,(x-1.4)+1.96,3.2\,(x-1.6)+2.56,3.6\,(x-1.8)+3.239999999999999,4\,(x-2)+4]$

(%o16) *done*

```
(%i17) p1: explicit(listoftangents, x, x0, xf)$
       p2: explicit(f(x), x, x0, xf)$
       wxdraw2d(
         nticks=200,
         color=blue,
         g,
         color=red,
         line_width=5,
         explicit(f(x),x,x0,xf),
         xlabel="x",
         ylabel="y");
```

(%t19)



(%o19)

6 Part V: Making Movies

Section 3.1

The following command generates a sequence of graphs and then saves them as an animated GIF file named "tanplot_anim.gif". This resulting file can typically be found in the following locations:

mac: in your applications folder
windows: C:\Program Files\Maxima-5.24.0\wxMaxima

Note that here we are using the draw( ) function instead of the draw2d( ) function. This is because of our need to create a graphic which is composed of several scenes (thus creating our animation). The draw2d( ) function is designed to create a graphic using only one scene. Thus draw2d(...) and draw(gr2d(...)) do the same thing.

```
(%i20) b: 1$
       tanplot: []$
       for a: 1 thru length(listoftangents) do block(
         tanplot: append(tanplot, [gr2d(
           nticks=200,
           color=blue,
           explicit(listoftangents[a],x,x0,xf),
           color=red,
           line_width=5,
           explicit(f(x),x,x0,xf),
           xlabel="x",
           ylabel="y",
           yrange=[-12,3])]))$
       draw(
         terminal = animated_gif,
         delay = 40,
         file_name  = "tanplot_anim",
         tanplot)$
```